ISE 307: Engineering Economic Analysis

Lecture 24: Software and Tools (Excel)

Term 253 (Summer)

Instructor: Mansur M. Arief

Industrial and Systems Engineering, KFUPM

Office: 22-219 — Email: mansur.arief@kfupm.edu.sa

Lecture 24

From Factor Tables to Formulas:
Every Method of This Course, in a Spreadsheet

Lecture 24 Objectives

1. Use the **five time-value functions** (FV, PV, PMT, RATE, NPER) with the **sign convention**
2. Build a **loan amortization schedule** that closes at exactly zero
3. Evaluate projects with **NPV** and **IRR** — and avoid the **year-zero trap**
4. Rebuild **depreciation and after-tax tables**: SLN, DDB, VDB, and a typed MACRS table
5. Automate what-if analysis: **Data Table**, **Goal Seek**, scenario rows, and the **AEC(N) table**

The Hand Method First, Then the Spreadsheet

- The factor method keeps the reasoning **visible**: the diagram shows *what* moves, the notation shows *how*, the lookup can be audited
- The spreadsheet automates the **arithmetic** — it will cheerfully compute the wrong model to fifteen digits
- Today's plan: rebuild problems you **already solved by hand**, so every sheet is checked against an answer you trust

The exam still requires the hand method: diagram, factor notation, substitution, boxed answer. The spreadsheet is for **checking** your hand work, and for the real projects after this course.

The Five Time-Value Functions

One function per unknown in the equivalence equation; all five share one argument family:

`rate`	interest rate per period, as a decimal
`nper`	number of periods N
`pmt`	equal payment A per period (0 if none)
`pv` / `fv`	present amount P / future amount F (0 if none)
`type`	0 for end-of-period flows (always 0 in this course)

`=FV(rate, nper, pmt, [pv])`	finds F	`=PV(rate, nper, pmt, [fv])`	finds P
`=PMT(rate, nper, pv, [fv])`	finds A	`=RATE(nper, pmt, pv, [fv])`	finds i
`=NPER(rate, pmt, pv, [fv])`	finds N		

One call can replace a two-factor hand solution — because `pv`, `pmt`, and `fv` can appear **together**.

The Sign Convention

In Excel's time-value functions, cash you **receive** is positive and cash you **pay out** is negative — every argument and every result.

The classic trap: the future worth of a 100,000 SAR deposit at 7% for 8 years.

formula	displays	reading
`=FV(0.07, 8, 0, 100000)`	`-171,818.62`	positive pv = a *loan*; result = repayment owed
`=FV(0.07, 8, 0, -100000)`	`171,818.62`	deposit (money *out*); result = withdrawal at year 8

Decide the direction of each flow *first*, exactly as on a cash flow diagram: **down arrows negative, up arrows positive**. Don't patch minus signs until the answer "looks right."

Four Hand Results, Four Calls

Problems we solved by hand earlier in the term, one function call each:

hand solution	formula	displays	by hand
loan payment, $(A/P, 8\%, 5)$	`=PMT(0.08, 5, 50000)`	`-12,522.82`	12,523
sinking fund, $(A/F, 7\%, 8)$	`=PMT(0.07, 8, 0, 100000)`	`-9,746.78`	9,747
equivalence rate	`=RATE(12, 28000, -200000)`	`9.05%`	$\approx 9.05\%$
years to double at 12%	`=NPER(0.12, 0, -1, 2)`	`6.12`	6.12

- The (A/P)–(A/F) factor pair is **one function** with different arguments filled in
- In `RATE`, the signs must **disagree** (give up one amount to receive the other), or the function returns an error

The Loan Amortization Schedule

The conveyor loan (50,000 SAR, 8%, 5 years). Payment: `=PMT(B2, B3, -B1)` → `12,522.82`. Year-1 row, then **fill down**:

B8 payment `=B$4`	C8 interest `=B$2*E7`	D8 principal `=B8-C8`	E8 balance `=E7-D8`

	A	B	C	D	E
6	year	payment	interest	principal	balance
7	0				50,000.00
8	1	12,522.82	4,000.00	8,522.82	41,477.18
9	2	12,522.82	3,318.17	9,204.65	32,272.53
10	3	12,522.82	2,581.80	9,941.02	22,331.51
11	4	12,522.82	1,786.52	10,736.30	11,595.21
12	5	12,522.82	927.62	11,595.21	**0.00**

Checks: final balance exactly 0; principal sums to 50,000; interest sums to 12,614.11. Single rows without the table: `=IPMT()`, `=PPMT()`, `=CUMIPMT()`.

Project Evaluation: NPV and IRR

The retrofit ($-55{,}000$; $+16{,}000$ in years 1–4; $+24{,}000$ in year 5; MARR 12%), flows in `B3:B8` with year 0 in `B3`:

	A	B
1	MARR	`0.12`
3	0	`-55000`
4	1	`16000`
	⋮	
8	5	`24000`
9	PW	`=B3+NPV(B1,B4:B8)`
10	IRR	`=IRR(B3:B8)`

`B9` $\rightarrow$ `7,215.83` (hand: $+7{,}216$) $\Rightarrow$ **accept**;
`B10` $\rightarrow$ `16.88%` $> 12\%$ — same verdict.

`=NPV(B1, B3:B8)` returns `6,442.71` — the right answer divided by 1.12. `IRR` assumes equal spacing; a skipped year is an explicit **0**, never an empty cell.

The year-zero trap: `NPV` discounts the *first* cell of its range by one full period — its range must start at **year 1**, with the year-0 flow added outside: `=A0 + NPV(rate, A1:AN)`. `IRR` is the opposite: its range **includes** year 0.

The PW Profile and the Incremental IRR

Profile: one anchored formula against a rate column, `=$B$3+NPV(D3, $B$4:$B$8)`, filled down:

	D	E
3	10%	17,495.20
4	15%	6,792.58
5	20%	−2,029.06

`=IRR(B3:B8)` → **18.76%** (hand interpolation: 18.78%)

Incremental IRR: build B−A as a literal column of `=C2-B2` formulas, then `IRR` it:

	A	B	C	D
1	year	A	B	B−A
2	0	−2,000	−8,000	−6,000
3	1	3,600	11,200	7,600
4	IRR	80%	40%	**26.67%**

Never subtract two IRRs: the incremental rate is the **rate of a difference**, not a difference of rates. 26.67% > MARR ⇒ choose B, agreeing with PW.

Depreciation: SLN, DDB, and VDB

All three share the argument head `(cost, salvage, life, ...)`. The flotation cell ($I = 50{,}000$, $S = 5{,}000$, $N = 5$):

	A	B	C
5	year	`=SLN($B$1,$B$2,$B$3)`	`=DDB($B$1,$B$2,$B$3,A6)`
6	1	9,000.00	20,000.00
7	2	9,000.00	12,000.00
8	3	9,000.00	7,200.00
9	4	9,000.00	4,320.00
10	5	9,000.00	**1,480.00**

- `DDB` applies the **salvage floor** itself: year 5 is 1,480, not 2,592 — so the salvage must be *typed correctly*
- Low salvage? `=VDB(10000, 0, 5, A6-1, A6)` has the **switch to SL built in**: 4,000 / 2,400 / 1,440 / 1,080 / 1,080

MACRS: Type the Table, Anchor the Basis

There is **no built-in MACRS function** — MACRS *is* a published table. The bagging equipment (basis 25,000, 5-year class):

	A	B	C	D
1	basis	25000		
3	year	p_n	D_n `=$B$1*B4`	BV_n `=$B$1-SUM($C$4:C4)`
4	1	20.00%	5,000.00	20,000.00
5	2	32.00%	8,000.00	12,000.00
6	3	19.20%	4,800.00	7,200.00
7	4	11.52%	2,880.00	4,320.00
8	5	11.52%	2,880.00	1,440.00
9	6	5.76%	1,440.00	**0.00**

- **Absolute reference** `$B$1`: locked with dollar signs, it doesn't shift on fill-down; the half-anchored `$C$4:C4` *stretches*
- Checks: charges sum to the full basis; final book value exactly 0 (MACRS salvage = 0)

The After-Tax Table

The bagging line (Lecture 22, example 3): one row per year, one column per pipeline step, tax rate anchored in `$B$2`:

	A	B	C	D	E	F	G	H
4	*n*	*R*	*E*	*D*	TI	tax	NI	ATCF
5	1	50,000	20,000	12,000	18,000	4,500	13,500	25,500
	⋮ (years 2–4 identical; only B–D are typed)							

Row-5 formulas, filled down: `D5: =SLN(60000, 12000, 4); E5: =B5-C5-D5; F5: =$B$2*E5; G5: =E5-F5; H5: =G5+D5`.

Disposal as a formula: $15{,}000 - 0.25\,(3{,}000) = 14{,}250$; final column $-60{,}000$; $25{,}500 \times 3$; 39,750: `=B12+NPV(0.10, B13:B16)` $\rightarrow$ `30,564.51` (hand: $+30{,}565.20$ — factor rounding)

Because the tax rate, salvage, and revenue live in **input cells**, this sheet is ready for every what-if question — no surgery needed.

One-Way Sensitivity: The Data Table

The crushing-unit base sheet (Lecture 23): inputs in `B1:B6`, net `B7` `=B3−B4`, and `B8` `=−B5−PV(B1,B2,B7,B6)` → `8,757.19`.

- **Data Table** = Excel's what-if tool: re-evaluates a formula cell for each candidate input, in one operation
- Layout: candidate revenues in a column (`D3:D7`); the link `=B8` one cell up-and-right (`E2`)
- Menu: Data → What-If Analysis → Data Table; *column input cell* = `B3` (the revenue input)

	D	E
2		=B8
3	33,600	−23,085.41
4	37,800	−7,164.11
5	42,000	8,757.19
6	46,200	24,678.50
7	50,400	40,599.80

The revenue row of Lecture 23's sensitivity table, to the SAR. Repeat per input, chart the columns ⇒ the sensitivity graph.

Break-Even with Goal Seek; Scenarios as Rows

Goal Seek adjusts one input until a formula cell reaches a stated value — exactly the $PW = 0$ solve:

Set cell: `B8` (PW)	To value: `0`	By changing: `B3` (revenue)	$\rightarrow R^* =$ `39,689.87`

So $A^* = 13{,}689.87$ (hand: 13,690); cushions 5.5% on revenue, 14.4% on the net. **Caution:** Goal Seek *overwrites* the input — read, then undo.

Scenarios change several inputs at once $\Rightarrow$ one row per coherent story, PW from that row's own inputs:

	A	B	C	D	E
11	scenario	revenue	op. cost	salvage	PW
12	pessimistic	36,000	28,000	3,000	−22,810.94
13	most likely	42,000	26,000	5,000	8,757.19
14	optimistic	46,000	24,000	7,000	32,743.76

Replacement: The AEC(N) Table in Two PMT Calls

The challenger truck (I = 72,000 in `B1`, i = 10% in `B2`). Each row: $AEC(N) = CR +$ operating AE:

	A	B	C	D	E	F
4	N	S_N	OC_N	CR	op. AE	$AEC(N)$
5	1	54,000	8,000	25,200.00	8,000.00	33,200.00
6	2	42,000	11,000	21,485.71	9,428.57	30,914.29
7	3	33,000	15,000	18,982.48	11,111.78	**30,094.26**
8	4	26,000	20,000	17,111.66	13,026.93	30,138.59

- CR: `=PMT($B$2, A5, -$B$1, B5)` — pay I now, recover S_N at the end: exactly $(I - S_N)(A/P) + iS_N$
- op. AE: `=PMT($B$2, A5, -NPV($B$2, $C$5:C5))` — discount the OC stream, spread it back over N
- Flag the bottom: `=MIN(F5:F8)` $\rightarrow$ 30,094.26; `=MATCH(MIN(...), F5:F8, 0)` $\rightarrow$ $N^* = 3$

Good Spreadsheet Habits

- **Inputs live in their own block** — labeled, one estimate per cell; everything below is formulas
- **No hard-coded numbers inside formulas**: `=$B$2*E7`, never `=0.08*E7`
- **Label the units**: "SAR/yr," "years," "fraction per year" — a rate typed as 8 where `0.08` is expected is half the errors
- **One change propagates everywhere**: change the MARR cell and every PW, profile point, and AEC row must update together
- **Check against a known answer** first — then edit the inputs toward the real problem

Tables round factors to 4 decimals; Excel carries 15. Last-digit gaps (a few SAR) are **expected**; large gaps mean a **sign error or a misplaced year**.

Exam-Style Spreadsheet Problems

These problems come **printed on paper** — no computer. Three formats:

1. **Fill the cell**: a sheet with blanked values — compute them by hand, using the pipeline logic the sheet encodes
2. **Write the formula**: state what belongs in a named cell — function, arguments, **ranges**, **anchors**, **signs**
3. **Interpret the output**: given a finished Data Table or Goal Seek result, deliver the verdict in **one sentence that carries the numbers**

Expect this format on the **final exam**. The sheet is the problem statement; the four steps (setup, formula, substitution, boxed answer) *plus the interpretation sentence* earn the marks.

Exam-Style Problem 1: Complete the After-Tax Sheet

A packaging machine: cost basis **80,000 SAR**, salvage **10,000**, life **5 years**, **straight line**; revenue **40,000**/yr; cash expenses **18,000**/yr; $t = 25\%$. Inputs block: basis `B1`, salvage `B2`, life `B3`, tax rate `B4`. The printed year-1 row:

	A	B	C	D	E	F	G	H
6	*n*	*R*	*E*	*D*	TI	tax	NI	ATCF
7	1	40,000	18,000	**(a)**	**(b)**	**(c)**	6,000	**(d)**

(a)–(d) Fill the four blanked cells.

(e) Write the formulas that belong in `D7` (depreciation) and `H7` (ATCF).

Exam-Style Problem 1: Solution

(a) $D = (80{,}000 - 10{,}000)/5 = \boxed{14{,}000}$ `D7: =SLN($B$1, $B$2, $B$3)`

(b) $TI = 40{,}000 - 18{,}000 - 14{,}000 = \boxed{8{,}000}$ **(c)** $\text{tax} = 0.25\,(8{,}000) = \boxed{2{,}000}$

Free check from the printed sheet: (b) $-$ (c) must reproduce the given NI: $8{,}000 - 2{,}000 = 6{,}000$ ✓

(d) $\text{ATCF} = \text{NI} + D = 6{,}000 + 14{,}000 = \boxed{20{,}000}$ `H7: =G7+D7`

Closed-form check: $(R-E)(1-t)+tD = 22{,}000\,(0.75)+0.25\,(14{,}000) = 16{,}500+3{,}500 = 20{,}000$ ✓

Exam-Style Problem 2: Read the Sensitivity Output

The machine as a project: invest **80,000**; ATCF **20,000**/yr (Problem 1); salvage **10,000** at year 5 (sold *at book value* $\Rightarrow$ no gains tax); MARR 10%:

$$PW = -80{,}000 + 20{,}000\,(3.7908) + 10{,}000\,(0.6209) = +2{,}025$$

PW (SAR)	-20%	-10%	base	$+10\%$	$+20\%$
annual revenue	**(a)**	$-9{,}347$	2,025	13,397	24,770
operating cost	12,260	7,143	2,025	**(b)**	$-8{,}210$

(a)–(b) Fill the two cells. *Hint:* each 1% of revenue moves the ATCF by $0.75 \times 400 = 300$, hence the PW by $300\,(3.7908) \approx 1{,}137$.

(c) Which input is more critical, and why? **(d)** Does the accept verdict survive a 10% revenue shortfall?

(e) Which Excel tool finds the break-even revenue *exactly* — and what are its three dialog

Exam-Style Problem 2: Solution

(a) $\Delta R = -8{,}000 \Rightarrow \Delta\text{ATCF} = -6{,}000$: $PW = 2{,}025 - 6{,}000\,(3.7908) = \boxed{-20{,}720}$

(b) $\Delta E = +1{,}800 \Rightarrow \Delta\text{ATCF} = -1{,}350$: $PW = 2{,}025 - 1{,}350\,(3.7908) = \boxed{-3{,}093}$

(c) Revenue: its PW swing is 45,490 vs. 20,470 for cost — the steepest line on the graph

(d) No: at -10% revenue, $PW = -9{,}347 < 0$. The cushion is only $2{,}025/1{,}137 \approx$ **1.8%**

(e) Goal Seek: Set cell = PW; To value = 0; By changing = the revenue cell $\rightarrow$ $R^* = 39{,}288$

The verdict sentence (it carries its own marks): "Accept at base, but the revenue cushion is under 2% — firm up the revenue forecast before committing."

Quick Check

A project's cash flows sit in `A1:A6`, with the year-0 investment in `A1`. Which formula gives the correct PW at 10%?

(A) `=NPV(0.10, A1:A6)`

(B) `=A1 + NPV(0.10, A2:A6)`

(C) `=NPV(0.10, A2:A6) - A1`

(D) `=IRR(A1:A6)`

Answer: (B). (A) slides every flow, including the investment, one year into the future — the result is the right answer divided by 1.10. (C) subtracts a *negative* investment twice. (D) returns a rate, not a worth.

Summary

- **Five functions**, one argument family, one **sign convention**: money in positive, money out negative — enter directions as on the CFD
- **Loan schedule**: `PMT` + three filled-down formulas; the final balance must be **exactly zero**
- `NPV` starts at year 1: `=A0+NPV(rate, A1:AN)`; `IRR` **includes** year 0; incremental IRR = `IRR` of a **difference column**
- `SLN` / `DDB` / `VDB` have the floor and the switch **built in**; MACRS is a **typed table** with absolute references; the ATCF sheet is the pipeline, one column per step
- **Data Table** = one-way sensitivity; **Goal Seek** = break-even; scenarios = rows; AEC(N) = **two PMT calls** + `MIN`/`MATCH`
- Inputs-first, no hard-coded numbers, check against a **known answer**; last-digit gaps vs. the tables are rounding, not errors

Before Class

To do before our class meeting:

1. **Read the Software and Tools chapter** (five functions through the AEC table, plus good habits) and open companion workbooks **01–04**
2. Complete the **Lecture 24 pre-class quiz** on Blackboard before class
3. Warm-up: predict what `=NPV(0.10, A1:A6)` returns when `A1` holds the year-0 investment — then write the corrected formula

In class (ICA-24): the economic-life AEC table and a break-even Goal Seek, built live — **bring a laptop** with Excel or Sheets.